
       

THE ALL-PAIRS PIPELINE

An all-pairs problem is a computation on every possible subset consisting of two elements

chosen from a set of n elements. N-body simulation and Householder reduction are all-

pairs problems. This chapter defines the all-pairs problem concisely by means of precedence

matrices and derives a parallel algorithm. The algorithm is presented in both coarse-grain

and medium-grain form. The all-pairs paradigm is illustrated by a pipeline for Householder

reduction of a matrix to triangular form.

1 INTRODUCTION

Successful exploitation of parallel computers depends to a large extent on
the development of useful concepts which enable programmers to view dif-
ferent applications as variations of a common theme. Our most fundamental
concepts, such as parallel processes and message communication, are embed-
ded in programming languages. In other cases, we discover programming
paradigms which can be used to solve a class of applications.

An all-pairs problem is a computation on every possible subset consisting
of two elements chosen from a set of n elements. N -body simulation is an
all-pairs problem (Fox 1988, Brinch Hansen 1991). Householder reduction
of a matrix to triangular form is a less obvious example (Press 1989, Brinch
Hansen 1990a). This paper develops the all-pairs paradigm discussed by
Shih (1987) and Cosnard (1988). I define the problem concisely by means of
precedence matrices and derive a parallel algorithm. The algorithm is pre-
sented in both coarse-grain and medium-grain form. The all-pairs paradigm
is illustrated by a pipeline for Householder reduction.

Pipeline algorithms for matrix reduction have already been developed
based on a detailed understanding of various reduction methods, such as
Gaussian elimination, Givens rotations, and Householder reduction (Ortega
1988).

P. Brinch Hansen, Studies in Computational Science, Chapter 4: The all-pairs pipeline.
Prentice Hall, Englewood Cliffs, NJ, (March 1995), 73–89. Copyright c© 2001, Per Brinch
Hansen.

I will take a different approach. I am convinced that the emphasis on
paradigms is the appropriate way to study parallel algorithms. I will illus-
trate the benefits of this approach by developing a parallel algorithm for
Householder reduction from a sequential algorithm. The program transfor-
mation is completely mechanical.

2 THE ALL-PROBLEMS PROBLEM

Let A be a set of n elements:

A = {a1, a2, . . . , an}

There are (n−1)n/2 ways to select a subset of A consisting of two elements:

{a2, a1}
{a3, a1} {a3, a2}
{a4, a1} {a4, a2} {a4, a3}
· · · · · · · · · · · ·

{an, a1} {an, a2} {an, a3} · · · {an, an−1}

Each subset {ai, aj} can be represented by an ordered pair (ai, aj), where
ai and aj are elements of A, and 1 ≤ j < i ≤ n.

An all-pairs computation performs an operation Q(ai, aj) on every pair
(ai, aj). This operation transforms ai and aj without involving any other
elements of A. Inspired by the n-body problem, I will say that the operation
defines an “interaction” between a pair of elements.

I will consider the all-pairs computation defined by Fig. 1. In this prece-
dence graph, an arrow from one operation to another indicates that the
former operation must be performed before the latter in any solution to the
problem. The figure shows that control flows from top to bottom and left
to right.

Element a1 interacts with a2, a3, . . . , an in that order. Element a2 in-
teracts with a1, a3, . . . , an, and so on. Finally, element an interacts with
a1, a2, . . . , an−1. All operations on a particular element ai take place strictly
one at a time. There is no possibility of race conditions when the all-pairs
computation is performed in parallel.

Figure 2 is a more compact representation of the precedence graph in
the form of a triangular precedence matrix. The elements of the precedence
matrix are operations. Each operation is preceded by the operations (if any)
immediately above and to the left of it, and is followed by the operations (if

Q(a2, a1)
↓

Q(a3, a1) → Q(a3, a2)
↓ ↓

Q(a4, a1) → Q(a4, a2) → Q(a4, a3)
↓ ↓ ↓
· · · · · · · · · · · · · · · · · · · · ·
↓ ↓ ↓

Q(an, a1) → Q(an, a2) → Q(an, a3) → · · · → Q(an, an−1)

Figure 1: All-pairs precedence graph.

any) immediately below and to the right of it. In other words, Q(ai, aj) is
preceded by Q(ai−1, aj) and Q(ai, aj−1) and is followed by Q(ai+1, aj) and
Q(ai, aj+1).

Q(a2, a1)
Q(a3, a1) Q(a3, a2)
Q(a4, a1) Q(a4, a2) Q(a4, a3)
· · · · · · · · · · · ·

Q(an, a1) Q(an, a2) Q(an, a3) · · · Q(an, an−1)

Figure 2: All-pairs precedence matrix.

3 SEQUENTIAL ALGORITHM

Figure 3 defines a sequential solution of the all-pairs problem for n elements
of type T . The correctness of the algorithm is obvious when you compare
it with Fig. 2. It defines the same sequence of operations as the precedence
matrix, column by column, from left to right.

Example 1:

An n-body simulation computes the trajectories of n particles which in-
teract through gravitational forces only. For each time step, the algorithm
computes the forces between each pair of particles (ai, aj) and adds them

type table = array [1..n] of T;
var a: table; i, j: integer;
for i := 1 to n − 1 do

for j := i + 1 to n do Q(a[j], a[i])

Figure 3: All-pairs algorithm.

to the total forces acting on these particles. The main loop of the force
summation is programmed as follows:

type system = array [1..n] of body;
var a: system; i, j: integer;
for i := 1 to n − 1 do

for j := i + 1 to n do
[sic] addforces(a[j], a[i])

Force interactions are symmetric, since addforces(aj , ai) is equivalent to
addforces(ai, aj). The example shows that an interaction between a pair
of elements may transform both elements. For large n, the O(nlogn) force
calculation of Barnes and Hut (1986) is much faster than the all-pairs algo-
rithm.

Example 2:

Gaussian elimination reduces an n×n real matrix to upper triangular
form in n−1 steps. In the ith step, the algorithm subtracts row ai multiplied
by aji/aii from row aj . If you ignore the (serious) rounding problems which
occur when the pivot element aii is very small, you have the following loop:

type matrix = array [1..n] of row;
var a: matrix; i, j: integer;
for i := 1 to n − 1 do

for j := i + 1 to n do
subtract(i, a[j], a[i])

The row interactions are asymmetric: subtract(i, aj , ai) is not the same as
subtract(j, ai, aj). Gaussian elimination without pivoting is numerically un-
stable (Press 1989). I use it only as a simple example of the all-pairs problem.
Householder reduction, which will be discussed later, is numerically stable
and well-suited for parallel execution.

Another sequential algorithm for the all-pairs problem is obtained by im-
plementing the precedence matrix, row by row, from top to bottom (Fig. 4).
For i = 1, the inner for statement defines an empty operation, so it makes
no difference whether the initial value of i is 1 or 2.

var a: table; i, j: integer;
for i := 1 to n do

for j := 1 to i − 1 do Q(a[i], a[j])

Figure 4: Equivalent all-pairs algorithm.

4 A COARSE-GRAIN PIPELINE

I will solve the all-pairs problem on a pipeline with p nodes, where 1 ≤ p ≤
n − 1 (Fig. 5). The nodes communicate by messages only. The first node
inputs the original elements of A. The last node outputs the final elements
of A. Without loss of generality, I assume that n− 1 is divisible by p. Each
node implements (n− 1)/p columns of the precedence matrix (Fig. 2).

Figure 5: The all-pairs pipeline.

The pipeline can be designed to output the elements in either natural
order a1, a2, . . . , an, or reverse order an, an−1, . . . , a1. I will use reverse output
to facilitate back substitution after matrix reduction.

I will program the pipeline nodes in SuperPascal. Each node has an
input channel, named left, and an output channel, named right. In program
assertions, a channel name denotes the sequence of messages transmitted
through the channel up to that point. As an example, the assertion

left = < ar..an >rev< a1..ar−1 >

shows that a node has input the elements ar through an, in that order,
followed by the elements a1 through ar−1 in reverse order. In other words,

left = < ar, ar+1, . . . , an, ar−1, ar−2, . . . , a1 >

Some sequences are, per definition, empty:

< ai..aj > = <>, rev< ai..aj > = <> for i > j

Figure 6 shows how the precedence matrix in Fig. 2 is partitioned for an
all-pairs pipeline with 2 nodes and 5 elements. An arrow in row i denotes
either input of element ai by the first node, communication of ai from the
first to the second node, or output of ai by the second node. At the end of
the computation, node 1 holds elements a1 and a2, node 2 stores a3 and a4,
while a5 has been output. The final task of the nodes is to output the stored
elements in reverse order a4, a3, a2, a1.

Figure 6: Precedence matrix of a pipeline.

Figure 7 shows the precedence matrix of a pipeline node that implements
columns r through s of Fig. 2, where 1 ≤ r ≤ s ≤ n−1. This matrix enables
you to develop an algorithm for a pipeline node.

A pipeline node goes through four phases:

1. Input phase: The node inputs elements ar through as and stores them
in a local array a. Every input element ai interacts with each of the
previously stored elements ar through ai−1.

{ left = <>, right = <> }
for i := r to s do

begin
receive(left, a[i]);
for j := r to i − 1 do Q(a[i], a[j])

end
{ left = < ar..as >, right = <> }

ar →
ar+1 → Q(ar+1, ar)
· · · · · · · · · · · ·
as → Q(as, ar) · · · Q(as, as−1)
as+1 → Q(as+1, ar) · · · Q(as+1, as−1) Q(as+1, as) → as+1

· · · · · · · · · · · · · · · · · · · · · · · ·
an → Q(an, ar) · · · Q(an, as−1) Q(an, as) → an

Figure 7: Precedence matrix of a pipeline node.

2. Transfer phase: The node inputs elements as+1 through an. Every
transfer element aj interacts with every local element and is then im-
mediately output to the next node. There is no room for transfer
elements in the local array. They are stored temporarily in a local
variable aj . (The last node transfers element an only, since s = n− 1.)
This phase completes the local computation defined by Fig. 7.

{ left = < ar..as >, right = <> }
for j := s + 1 to n do

begin
receive(left, aj);
for i := r to s do Q(aj, a[i]);
send(right, aj)

end
{ left = < ar..an >, right = < as+1..an > }

3. Output phase: The node outputs the local elements in reverse order.

{ left = < ar..an >, right = < as+1..an > }
for i := s downto r do

send(right, a[i])
{ left = < ar..an >,

right = < as+1..an >rev< ar..as > }

4. Copy phase: The node copies all elements output in reverse order by
the previous nodes. (The first node copies no elements since r = 1.)

{ left = < ar..an >,
right = < as+1..an >rev< ar..as > }
for j := r − 1 downto 1 do

begin
receive(left, aj);
send(right, aj)

end
{ left = < ar..an >rev< a1..ar−1 >,

right = < as+1..an >rev< a1..as > }

Putting these program pieces together, I obtain the complete algorithm
for a pipeline node (Fig. 8). To suppress irrelevant detail, I use an array
type with dynamic bounds r..s (which does not exist in SuperPascal).

The algorithm does not duplicate the whole set A within each node. The
first n− 1 elements of the set are distributed evenly among the nodes of the
pipeline. The last element is transferred through the pipeline without being
stored.

The postcondition of the last phase shows that the input sequence of
a node is a function of its lower bound r, while the output sequence is
determined by the upper bound s:

left(r) = < ar..an >rev< a1..ar−1 >
right(s) = < as+1..an >rev< a1..as >

This assertion implies that the first node inputs the numbers in natural
order:

left(1) = < a1..an >rev< a1..a0 > = < a1..an >

while the last node outputs them in reverse order:

right(n − 1) = < an..an >rev< a1..an−1 > = rev< a1..an >

I leave it as an exercise for you to write a modified algorithm which
accepts input and produces output in natural order. The key idea is to use
the input/output sequences

left(r) = < ar..an−1 >< a1..ar−1 >< an >
right(s) = < as+1..an−1 >< a1..as >< an >

procedure node(r, s: integer;
left, right: channel);

type block = array [r..s] of T;
var a: block; aj: T; i, j: integer;
begin
{ 1 <= r <= s <= n − 1 }
for i := r to s do

begin
receive(left, a[i]);
for j := r to i − 1 do Q(a[i], a[j])

end;
for j := s + 1 to n do

begin
receive(left, aj);
for i := r to s do Q(aj, a[i]);
send(right, aj)

end;
for i := s downto r do

send(right, a[i]);
for j := r − 1 downto 1 do

begin
receive(left, aj);
send(right, aj)

end
end;

Figure 8: Node algorithm.

The all-pairs paradigm enables a programmer to formulate parallel ver-
sions of similar sequential algorithms by trivial substitution.

Example 3:

You can derive a pipelined algorithm for the force summation in n-body
simulation by performing the following substitutions in Fig. 8:

type body replaces type T
addforces(a[i], a[j]) replaces Q(a[i], a[j])
addforces(aj, a[i]) replaces Q(aj, a[i])

By setting r = 1 and s = n − 1 in Fig. 8, you obtain a single-processor
version of the all-pairs pipeline which is equivalent to Fig. 4.

5 A MEDIUM-GRAIN PIPELINE

A medium-grain pipeline consists of n − 1 nodes, each of which holds only
one element of the set A. The medium-grain algorithm is derived from the
coarse-grain version by setting i = r = s in Fig. 8. Figure 9 defines a node
that implements the ith column of the precedence matrix (Fig. 2).

procedure node(i: integer;
left, right: channel);

var ai, aj: T; j: integer;
begin
{ 1 <= i <= n − 1 }
receive(left, ai);
for j := i + 1 to n do

begin
receive(left, aj);
Q(aj, ai);
send(right, aj)

end;
send(right, ai);
for j := i − 1 downto 1 do

begin
receive(left, aj);
send(right, aj)

end
end;

Figure 9: Medium-grain node algorithm.

Example 4:

From a sequential algorithm for Gaussian elimination without pivoting,
you can design a pipeline algorithm by making the following substitutions
in Fig. 9:

type row replaces type T
subtract(i, aj, ai) replaces Q(aj, ai)

6 VARIATION ON A THEME

In the all-pairs computation discussed so far, each operation is an interaction
between two elements of the same set

A = {a1, a2, . . . , an}

In some applications, it is more convenient to use A to compute another
set

B = {b1, b2, . . . , bn−1}

and let the elements of A interact with the elements of B. The set B is a
temporary data structure which exists during the computation only.

Figure 10 shows the precedence matrix for this variant of the all-pairs
computation.

P (a1, b1)
Q(a2, b1) P (a2, b2)
Q(a3, b1) Q(a3, b2) P (a3, b3)
Q(a4, b1) Q(a4, b2) Q(a4, b3)
· · · · · · · · · · · ·
Q(an−1, b1) Q(an−1, b2) Q(an−1, b3) · · · P (an−1, bn−1)
Q(an, b1) Q(an, b2) Q(an, b3) · · · Q(an, bn−1)

Figure 10: Variant precedence matrix.

The all-pairs variant is a computation on every set {ai, bj}, where ai is
a member of A, bj is a member of B, and j ≤ i. For each of these sets, one
of two operations is performed:

1. The operation P (ai, bi) transforms element ai and computes the cor-
responding element bi, where 1 ≤ i ≤ n− 1.

2. The operation Q(ai, bj) transforms elements ai and bj , where 1 ≤ j <
i ≤ n.

From the precedence matrix, I derive a sequential algorithm (Fig. 11).
In this case, each element of B exists only during a single step of the com-
putation. So, the set B is represented by a variable bi, which holds a single
element only. This is a variant of Fig. 3.

var a: table; bi: T; i, j: integer;
for i := 1 to n − 1 do

begin
P(a[i], bi);
for j := i + 1 to n do Q(a[j], bi)

end

Figure 11: All-pairs variant.

Example 5:

Householder’s method reduces an n×n real matrix to upper triangular
form in n− 1 steps. The main loop of a sequential Householder reduction is
shown below (Brinch Hansen 1990a):

type matrix = array [1..n] of column;
var a: matrix; vi: column; i, j: integer;
for i := 1 to n − 1 do

begin
eliminate(i, a[i], vi);
for j := i + 1 to n do

transform(i, a[j], vi)
end

The matrix is stored by columns, that is, a[i] denotes the ith column
of A. In the ith step, the algorithm uses column a[i] to compute a column
vector vi. This vector is then used to transform each remaining column a[j],
where i+ 1 ≤ j ≤ n. The eliminate and transform operations are defined in
Case Study 10. The elements of the set A are matrix columns a1 through
an. The elements of the set B are column vectors v1 through vn−1. For
each element ai of A (except an), the algorithm computes the corresponding
element vi of B.

Figure 12 is a variant of Fig. 4 obtained from Fig. 10.
Figure 13 defines a pipeline node for the all-pairs variant. All elements of

A and B (except an) are distributed evenly among the nodes. The elements
of B are temporary local entities which are not transmitted between the
nodes.

For a = b and P = empty, the algorithm reduces to the algorithm in
Fig. 8. A medium-grain version of this pipeline is similar to Fig. 9.

var a, b: table; i, j: integer;
for i := 1 to n − 1 do

begin
for j := 1 to i − 1 do Q(a[i], b[j]);
P(a[i], b[i])

end;
for i := 1 to n − 1 do Q(a[n], b[i])

Figure 12: Equivalent all-pairs variant.

7 AN EXAMPLE: HOUSEHOLDER REDUCTION

Many problems in science and engineering involve a system of n linear equa-
tions. The equations can be solved in two steps: First, the equations are
reduced to triangular form by systematic elimination of unknowns. The
triangular equations are then solved by back substitution.

The most time-consuming part of the computation is the reduction of the
coefficient matrix to triangular form. The standard Gaussian and Gauss-
Jordan eliminations are straightforward reduction algorithms. They do,
however, require pivoting, a rearrangement of the rows and columns, which
in most cases, prevents numerical instability (Press 1989). On a parallel
computer, pivoting complicates these algorithms (Fox 1988).

For a parallel computer, Householder reduction is an attractive method
that is numerically stable and does not require pivoting (Press 1989, Brinch
Hansen 1990a). In the following, I derive a pipeline algorithm for House-
holder reduction directly from the all-pairs paradigm.

Example 5 defines the main loop of sequential Householder reduction.
The theory behind Householder reduction is explained in Brinch Hansen
(1990a) and will not be repeated here.

A comparison of Fig. 11 and Example 5 shows that Householder reduc-
tion is an all-pairs variant. So, you can derive a pipeline for Householder
reduction by making the following substitutions in Fig. 13:

type column replaces type T
variable v replaces variable b
eliminate(i, a[i], v[i]) replaces P(a[i], b[i])
transform(j, a[i], v[j]) replaces Q(a[i], b[j])
transform(i, aj, v[i]) replaces Q(aj, b[i])

Figure 14 defines a node of the Householder pipeline which holds columns



   

14 PER BRINCH HANSEN

procedure node(r, s: integer;
left, right: channel);

type block = array [r..s] of T;
var a, b: block; aj: T; i, j: integer;
begin
{ 1 <= r <= s <= n − 1 }
for i := r to s do

begin
receive(left, a[i]);
for j := r to i − 1 do Q(a[i], b[j]);
P(a[i], b[i])

end;
for j := s + 1 to n do

begin
receive(left, aj);
for i := r to s do Q(aj, b[i]);
send(right, aj)

end;
for i := s downto r do

send(right, a[i]);
for j := r − 1 downto 1 do

begin
receive(left, aj);
send(right, aj)

end
end;

Figure 13: The all-pairs variant.

r through s, where 1 ≤ r ≤ s ≤ n − 1. The pipeline inputs the columns
in natural order, reduces the matrix to triangular form, and outputs the
final columns in reverse order. The performance of the parallel algorithm
has been analyzed and measured on a Computing Surface (Brinch Hansen
1990b).

The parallel Householder algorithm is an ideal algorithm for experiment-
ing with a parallel computer:

1. It is a fundamental algorithm of considerable practical value.

procedure node(r, s: integer;
left, right: channel);

type block = array [r..s] of column;
var a, v: block; aj: column;

i, j: integer;
begin
{ 1 <= r <= s <= n − 1 }
for i := r to s do

begin
receive(left, a[i]);
for j := r to i − 1 do

transform(j, a[i], v[j]);
eliminate(i, a[i], v[i])

end;
for j := s + 1 to n do

begin
receive(left, aj);
for i := r to s do

transform(i, aj, v[i]);
send(right, aj)

end;
for i := s downto r do

send(right, a[i]);
for j := r − 1 downto 1 do

begin
receive(left, aj);
send(right, aj)

end
end;

Figure 14: Householder node algorithm.

2. It demonstrates the use of a general paradigm to transform a sequential
algorithm into a parallel one.

3. It illustrates the subtleties of distributing a large computation evenly
among parallel processors.

8 FINAL REMARKS

After programming n-body simulation and Householder reduction in occam
for the Computing Surface, I was delighted to discover that these seem-
ingly unrelated problems can be solved by refinements of the same abstract
program.

I have presented pipeline algorithms for two variants of the all-pairs
paradigm. As a non-trivial example, I have used the paradigm to derive
a pipeline algorithm for Householder reduction of a real matrix to triangular
form. The parallel algorithm was derived from a sequential one by trivial
substitution of data types, variables, and procedure statements.

